
1. Overview

This document lists TTCN changes needed to correct issues in the ATS ‘iwd-TTCN3-B2015-03_D16wk10’.

3. Corrections
Change 1 – Correction to function ‘f_TC_11_2_8_GERAN’
	Function name
	f_TC_11_2_8_GERAN ()

	Reason for change
	TCH channel is created and initialized in the preamble of InterRAT TCs with IMS relation (EUTRAN to GERAN case). UE moves from EUTRAN cell to GERAN cell. Ciphering settings on TCH channel are not updated with the settings after authentication and ciphering procedures on GERAN side during MO call or Emergency Call setup. TCH channel is configured with incorrect ciphering parameters from initial TCH channel creation and assignment from SDCCH to TCH fails.

	Summary of change
	Move the TCH channel creation functions after authentication and ciphering procedures in MO call and Emergency Call setup procedures which ensures initialisation of TCH channel with matching ciphering settings.
Together power setting function for TCH channel needs to be moved after TCH channel creation. 

	TTCN module
	LTE_IRAT\11\IMS_EmergencyCall_UGC_GERAN.ttcn

	MCC160 Comment
	Not Required.  This is contrary to TTCN2 behaviour.


Before change

...
  /*

   * @desc      REFERENCE TS 36.523-1 clause 11.2.8

   * @status    APPROVED (LTE_IRAT)

   */

  function f_TC_11_2_8_GERAN() runs on GERAN_PTC

  {

    var G_LogicChType v_TchType;

    //Initialise cell info and configure cell 24

    f_GERAN_Init(EUTRA_GERAN);

    f_GERAN_CreateCell(geran_Cell24);

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    f_GERAN_SetCellPowerLevel(geran_Cell24, tsc_GERAN_SuitableNeighbourCell);

    //Receive updated auth parameters

    f_GERAN_InterRAT_InitialiseAuthParams();

    f_GERAN_TestBody_Set(true);

    //Create the physical channel in GERAN cell

    f_GERAN_CreateTCH(geran_Cell24);
    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    f_GERAN_SetCellPowerLevel(geran_Cell24, tsc_GERAN_ServingCell);

    f_GERAN_SetChPwrLevel(geran_Cell24, tsc_TchCh, tsc_GERAN_ServingCell);
    //@siclog "Step 4b1 - 4b7" siclog@

    v_TchType := f_GERAN_EmgCallSetup(geran_Cell24);

    f_GERAN_TestBody_Set(false);

    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger);     // Tell EUTRA test is over

    //Perform postamble

    f_GERAN_CC_Disconnect(geran_Cell24, v_TchType, -, cs_TI_MO); // @sic R5s150525 sic@

    f_GERAN_Postamble(geran_Cell24, G5_DEREGISTERED, -, -, v_TchType);  // @sic R5s150525 sic@

  } // f_TC_11_2_8_GERAN
...

After change

...

  /*

   * @desc      REFERENCE TS 36.523-1 clause 11.2.8

   * @status    APPROVED (LTE_IRAT)

   */

  function f_TC_11_2_8_GERAN() runs on GERAN_PTC

  {

    var G_LogicChType v_TchType;

    //Initialise cell info and configure cell 24

    f_GERAN_Init(EUTRA_GERAN);

    f_GERAN_CreateCell(geran_Cell24);

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    f_GERAN_SetCellPowerLevel(geran_Cell24, tsc_GERAN_SuitableNeighbourCell);

    //Receive updated auth parameters

    f_GERAN_InterRAT_InitialiseAuthParams();

    f_GERAN_TestBody_Set(true);

    //Create the physical channel in GERAN cell

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    f_GERAN_SetCellPowerLevel(geran_Cell24, tsc_GERAN_ServingCell);

    //@siclog "Step 4b1 - 4b7" siclog@

    v_TchType := f_GERAN_EmgCallSetup(geran_Cell24);

    f_GERAN_TestBody_Set(false);

    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger);     // Tell EUTRA test is over

    //Perform postamble

    f_GERAN_CC_Disconnect(geran_Cell24, v_TchType, -, cs_TI_MO); // @sic R5s150525 sic@

    f_GERAN_Postamble(geran_Cell24, G5_DEREGISTERED, -, -, v_TchType);  // @sic R5s150525 sic@

  } // f_TC_11_2_8_GERAN
... 
Change 2 – Correction to function ‘f_GERAN_EmgCallSetup’
	Function name
	f_GERAN_EmgCallSetup ()

	Reason for change
	See change 1.

	Summary of change
	Move the TCH channel creation functions after authentication and ciphering procedures in MO call and Emergency Call setup procedures which ensures initialisation of TCH channel with matching ciphering settings.
Together power setting function for TCH channel needs to be moved after TCH channel creation. 

	TTCN module
	Common/GERAN/GERAN_CommonFunctions.ttcn

	MCC160 Comment
	Not Required.  This is contrary to TTCN2 behaviour.


Before change

...
  //----------------------------------------------------------------------------

  /*

   * @desc      To get the UE into Call Active, U10 state, steps 5 to 19 of 51.010, 10.2.3

   * @param     p_CellId

   * @return    G_LogicChType        Logical channel of TCH UE is now on N.B. As a full rate TCH is configured in this step, the SubChannel parameter is not used

   * @status    APPROVED (IMS_IRAT, LTE_IRAT)

   */

  function f_GERAN_EmgCallSetup(GERAN_CellId_Type p_CellId) runs on GERAN_PTC return G_LogicChType

  {

    var G_L2_DATAMESSAGE_IND v_ServReq;

    var G_L2_DATAMESSAGE_IND v_SetUp;

    var SubChannelNumber v_SubChannel;

    var G_LogicChType v_ChannelType;

    var G_LogicChType v_TchType;

    var TI v_Ti_R;

    var RFN v_RFN;

    var integer v_RR_RA;

    var G_L2_DATAMESSAGE_IND v_ChanReq;

    var B10_Type v_BCCH := f_GERAN_BCCHFreq_Get(p_CellId);

    var B3_Type v_BCC := f_GERAN_BCC_Get(p_CellId);

    timer t_Wait60s := 60.0;

    timer t_Wait1s := 1.0; // simple timer to wait for a Suspension Req

    t_Wait60s.start;

    //Receive CHANNEL REQUEST

    alt {

      [] G_L2.receive(car_G_L2_ACCESS_IND(p_CellId,

                                          tsc_PhyCh0,

                                          tsc_RACH,

                                          ?,

                                          ?,

                                          cr_G_ChannelReq ('101?????'B))) // type emergency call

        -> value v_ChanReq

        {

          t_Wait60s.stop;

          v_RFN := v_ChanReq.access.rfn;

          v_RR_RA := bit2int(v_ChanReq.access.burst.chReq.estCauRandomRef);

          f_GERAN_PreliminaryPass(__FILE__, __LINE__, "Channel Request for Emergency Call Received");

        }

      [] t_Wait60s.timeout

        {

          f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "No channel request received");

        }

    }

    //Send IMMEDIATE ASSIGNMENT

    G_L2.send(cas_G_L2_UNITDATA_REQ(p_CellId,

                                    tsc_PhyCh0,

                                    tsc_AGCH,

                                    15,

                                    cs_G_RFN_Omit,

                                    cs_ImmAss(cs_ImmediateAssignment(v_BCCH, v_BCC, v_RR_RA, v_RFN, substr(oct2bit(f_GERAN_TA_Get(p_CellId)), 1, 6))))); // @sic R5s120587 sic@

    //Receive CM Service Request

    G_L2.receive(car_G_L2_Estab_IND(geran_Cell24, ?, ?, ?, cr_G_CMServiceRequest(?))) -> value v_ServReq;

    f_GERAN_PreliminaryPass(__FILE__, __LINE__, "CM Service Request Received");

    v_ChannelType := v_ServReq.sabm.g_LogicChType;

    v_SubChannel := v_ServReq.sabm.subChannel;

    // !!!! Classmark Change messages are not included in the prose !!!!

    //Receive ClassmarkChange

    G_L2.receive(car_G_L2_DATA_IND(p_CellId, ?, tsc_PhyCh0, ?, ?, ?, cr_G_ClassmarkChangeAny));

    alt { // if UTRAN Classmark Change arrives first

      [pc_UTRA] G_L2.receive(car_G_L2_DATA_IND(p_CellId, ?, tsc_PhyCh0, ?, ?, ?, cr_G_UtranClassmarkChange))

        {

          t_Wait1s.start;

          alt { // Wait a while to see if a GPRS Suspension Req is also sent

            [] G_L2.receive(car_G_L2_DATA_IND(p_CellId, ?, tsc_PhyCh0, ?, ?, ?, cr_GPRS_SuspensionReq(?, cr_RAI, ?)))

              {

                t_Wait1s.stop;

              }

            [] t_Wait1s.timeout {}

          }

        }

      [] G_L2.receive(car_G_L2_DATA_IND(p_CellId, ?, tsc_PhyCh0, ?, ?, ?, cr_GPRS_SuspensionReq(?, cr_RAI, ?)))

        {

          if (pc_UTRA) {

            G_L2.receive(car_G_L2_DATA_IND(p_CellId, ?, tsc_PhyCh0, ?, ?, ?, cr_G_UtranClassmarkChange));

          }

        }

    }

    //AUTHENTICATION REQUEST and AUTHENTICATION RESPONSE

    f_GERAN_MM_Authentication(p_CellId, v_ChannelType, v_SubChannel);

    // !!!! Ciphering NOT INCLUDED in the prose !!!!

    f_GERAN_Ciphering_Mode_Setting (p_CellId, tsc_PhyCh0, v_ChannelType, v_SubChannel);

    G_L2.receive(car_G_L2_DATA_IND(p_CellId, ?, ?, ?, ?, ?,  cr_G_EmgSetup)) -> value v_SetUp;

    v_Ti_R := v_SetUp.l3.msg.esetup.ti;

    v_TchType :=  f_GERAN_EnterU10_MO_Steps11_19 (p_CellId, v_ChannelType, v_SubChannel, v_Ti_R);

    f_GERAN_PreliminaryPass(__FILE__, __LINE__, "Emergency Call Setup Received");

    return v_TchType;

  }
...

After change

...

  //----------------------------------------------------------------------------

  /*

   * @desc      To get the UE into Call Active, U10 state, steps 5 to 19 of 51.010, 10.2.3

   * @param     p_CellId

   * @return    G_LogicChType        Logical channel of TCH UE is now on N.B. As a full rate TCH is configured in this step, the SubChannel parameter is not used

   * @status    APPROVED (IMS_IRAT, LTE_IRAT)

   */

  function f_GERAN_EmgCallSetup(GERAN_CellId_Type p_CellId) runs on GERAN_PTC return G_LogicChType

  {

    var G_L2_DATAMESSAGE_IND v_ServReq;

    var G_L2_DATAMESSAGE_IND v_SetUp;

    var SubChannelNumber v_SubChannel;

    var G_LogicChType v_ChannelType;

    var G_LogicChType v_TchType;

    var TI v_Ti_R;

    var RFN v_RFN;

    var integer v_RR_RA;

    var G_L2_DATAMESSAGE_IND v_ChanReq;

    var B10_Type v_BCCH := f_GERAN_BCCHFreq_Get(p_CellId);

    var B3_Type v_BCC := f_GERAN_BCC_Get(p_CellId);

    timer t_Wait60s := 60.0;

    timer t_Wait1s := 1.0; // simple timer to wait for a Suspension Req

    t_Wait60s.start;

    //Receive CHANNEL REQUEST

    alt {

      [] G_L2.receive(car_G_L2_ACCESS_IND(p_CellId,

                                          tsc_PhyCh0,

                                          tsc_RACH,

                                          ?,

                                          ?,

                                          cr_G_ChannelReq ('101?????'B))) // type emergency call

        -> value v_ChanReq

        {

          t_Wait60s.stop;

          v_RFN := v_ChanReq.access.rfn;

          v_RR_RA := bit2int(v_ChanReq.access.burst.chReq.estCauRandomRef);

          f_GERAN_PreliminaryPass(__FILE__, __LINE__, "Channel Request for Emergency Call Received");

        }

      [] t_Wait60s.timeout

        {

          f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "No channel request received");

        }

    }

    //Send IMMEDIATE ASSIGNMENT

    G_L2.send(cas_G_L2_UNITDATA_REQ(p_CellId,

                                    tsc_PhyCh0,

                                    tsc_AGCH,

                                    15,

                                    cs_G_RFN_Omit,

                                    cs_ImmAss(cs_ImmediateAssignment(v_BCCH, v_BCC, v_RR_RA, v_RFN, substr(oct2bit(f_GERAN_TA_Get(p_CellId)), 1, 6))))); // @sic R5s120587 sic@

    //Receive CM Service Request

    G_L2.receive(car_G_L2_Estab_IND(geran_Cell24, ?, ?, ?, cr_G_CMServiceRequest(?))) -> value v_ServReq;

    f_GERAN_PreliminaryPass(__FILE__, __LINE__, "CM Service Request Received");

    v_ChannelType := v_ServReq.sabm.g_LogicChType;

    v_SubChannel := v_ServReq.sabm.subChannel;

    // !!!! Classmark Change messages are not included in the prose !!!!

    //Receive ClassmarkChange

    G_L2.receive(car_G_L2_DATA_IND(p_CellId, ?, tsc_PhyCh0, ?, ?, ?, cr_G_ClassmarkChangeAny));

    alt { // if UTRAN Classmark Change arrives first

      [pc_UTRA] G_L2.receive(car_G_L2_DATA_IND(p_CellId, ?, tsc_PhyCh0, ?, ?, ?, cr_G_UtranClassmarkChange))

        {

          t_Wait1s.start;

          alt { // Wait a while to see if a GPRS Suspension Req is also sent

            [] G_L2.receive(car_G_L2_DATA_IND(p_CellId, ?, tsc_PhyCh0, ?, ?, ?, cr_GPRS_SuspensionReq(?, cr_RAI, ?)))

              {

                t_Wait1s.stop;

              }

            [] t_Wait1s.timeout {}

          }

        }

      [] G_L2.receive(car_G_L2_DATA_IND(p_CellId, ?, tsc_PhyCh0, ?, ?, ?, cr_GPRS_SuspensionReq(?, cr_RAI, ?)))

        {

          if (pc_UTRA) {

            G_L2.receive(car_G_L2_DATA_IND(p_CellId, ?, tsc_PhyCh0, ?, ?, ?, cr_G_UtranClassmarkChange));

          }

        }

    }

    //AUTHENTICATION REQUEST and AUTHENTICATION RESPONSE

    f_GERAN_MM_Authentication(p_CellId, v_ChannelType, v_SubChannel);

    // !!!! Ciphering NOT INCLUDED in the prose !!!!

    f_GERAN_Ciphering_Mode_Setting (p_CellId, tsc_PhyCh0, v_ChannelType, v_SubChannel);

    // Create TCH channel with initialized ciphering parameters

    f_GERAN_CreateTCH(p_CellId);

    f_GERAN_SetChPwrLevel(p_CellId, tsc_TchCh, tsc_GERAN_ServingCell);
    G_L2.receive(car_G_L2_DATA_IND(p_CellId, ?, ?, ?, ?, ?,  cr_G_EmgSetup)) -> value v_SetUp;

    v_Ti_R := v_SetUp.l3.msg.esetup.ti;

    v_TchType :=  f_GERAN_EnterU10_MO_Steps11_19 (p_CellId, v_ChannelType, v_SubChannel, v_Ti_R);

    f_GERAN_PreliminaryPass(__FILE__, __LINE__, "Emergency Call Setup Received");

    return v_TchType;

  }
... 
Change 3 – Correction to function ‘f_GERAN_EnterU10_MO_Steps5_19’
	Function name
	f_GERAN_EnterU10_MO_Steps5_19 ()

	Reason for change
	See change 1.

	Summary of change
	Move the TCH channel creation functions after Authentication and Ciphering procedures in MO call and Emergency call setup procedures which ensures initialisation of TCH channel with matching ciphering settings.
Together power setting function for TCH channel needs to be moved after TCH channel creation. 

	TTCN module
	Common/GERAN/GERAN_CommonFunctions.ttcn

	MCC160 Comment
	Not Required.  This is contrary to TTCN2 behaviour.


Before change

...
  //----------------------------------------------------------------------------

  /*

   * @desc      To get the UE into Call Active, U10 state, steps 5 to 19 of 51.010, 10.2.3

   * @param     p_CellId

   * @param     p_ChannelType        Logical channel of SDCCH UE is currently on

   * @param     p_SubChannel         subchannel of SDCCH UE is currently on

   * @return    G_LogicChType        Logical channel of TCH UE is now on N.B. As a full rate TCH is configured in this step, the SubChannel parameter is not used

   * @param     p_StopAtU4State     (default value: false) to stop after Alerting message @sic R5-150427 sic@

   * @status    APPROVED (IMS_IRAT, LTE_IRAT)

   */

  function f_GERAN_EnterU10_MO_Steps5_19(GERAN_CellId_Type p_CellId,

                                         G_LogicChType     p_ChannelType,

                                         SubChannelNumber  p_SubChannel,

                                          boolean           p_StopAtU4State := false) runs on GERAN_PTC return G_LogicChType

  {

    var G_L2_DATAMESSAGE_IND v_SetUp;

    var TI v_Ti_R; //Receiving Transaction Identifier

    //Step 5-6

    //AUTHENTICATION REQUEST and AUTHENTICATION RESPONSE

    f_GERAN_MM_Authentication(p_CellId, p_ChannelType, p_SubChannel);

    //Step 7-9

    //CIPHERING MODE COMMAND & CIPHERING MODE COMPLETE

    f_GERAN_Ciphering_Mode_Setting(p_CellId, tsc_PhyCh0, p_ChannelType, p_SubChannel);

    //Step 10

    //Receive Set Up message

    G_L2.receive(car_G_L2_DATA_IND(p_CellId, ?, ?, ?, ?, ?,  cr_G_SetupUL_MO)) -> value v_SetUp;

    v_Ti_R := v_SetUp.l3.msg.setupul.ti;

    return f_GERAN_EnterU10_MO_Steps11_19 (p_CellId, p_ChannelType, p_SubChannel, v_Ti_R, p_StopAtU4State); // @sic R5-142954 sic@

  }
...

After change

...

  //----------------------------------------------------------------------------

  /*

   * @desc      To get the UE into Call Active, U10 state, steps 5 to 19 of 51.010, 10.2.3

   * @param     p_CellId

   * @param     p_ChannelType        Logical channel of SDCCH UE is currently on

   * @param     p_SubChannel         subchannel of SDCCH UE is currently on

   * @return    G_LogicChType        Logical channel of TCH UE is now on N.B. As a full rate TCH is configured in this step, the SubChannel parameter is not used

   * @param     p_StopAtU4State     (default value: false) to stop after Alerting message @sic R5-150427 sic@

   * @status    APPROVED (IMS_IRAT, LTE_IRAT)

   */

  function f_GERAN_EnterU10_MO_Steps5_19(GERAN_CellId_Type p_CellId,

                                         G_LogicChType     p_ChannelType,

                                         SubChannelNumber  p_SubChannel,

                                          boolean           p_StopAtU4State := false) runs on GERAN_PTC return G_LogicChType

  {

    var G_L2_DATAMESSAGE_IND v_SetUp;

    var TI v_Ti_R; //Receiving Transaction Identifier

    //Step 5-6

    //AUTHENTICATION REQUEST and AUTHENTICATION RESPONSE

    f_GERAN_MM_Authentication(p_CellId, p_ChannelType, p_SubChannel);

    //Step 7-9

    //CIPHERING MODE COMMAND & CIPHERING MODE COMPLETE

    f_GERAN_Ciphering_Mode_Setting(p_CellId, tsc_PhyCh0, p_ChannelType, p_SubChannel);

    // Create TCH channel with initialized ciphering parameters

    f_GERAN_CreateTCH(p_CellId);

    f_GERAN_SetChPwrLevel(p_CellId, tsc_TchCh, tsc_GERAN_ServingCell);
    //Step 10

    //Receive Set Up message

    G_L2.receive(car_G_L2_DATA_IND(p_CellId, ?, ?, ?, ?, ?,  cr_G_SetupUL_MO)) -> value v_SetUp;

    v_Ti_R := v_SetUp.l3.msg.setupul.ti;

    return f_GERAN_EnterU10_MO_Steps11_19 (p_CellId, p_ChannelType, p_SubChannel, v_Ti_R, p_StopAtU4State); // @sic R5-142954 sic@

  }
... 
Change 4 – Alignment of functions f_TC_13_1_8_GERAN, f_TC_13_1_9_GERAN due to common changes
	Function name
	f_TC_13_1_8_GERAN ()
f_TC_13_1_9_GERAN ()

	Reason for change
	See change 1.

	Summary of change
	Move the TCH channel creation functions after Authentication and Ciphering procedures in MO call and Emergency call setup procedures which ensures initialisation of TCH channel with matching ciphering settings.
Together power setting function for TCH channel needs to be moved after TCH channel creation.
EUTRAN-GERAN Multi-layer TCs 13.1.8, 13.1.9 need to be adjusted because TCH creation has been moved to MO call function  f_GERAN_EnterU10_MO_Steps5_19.



	TTCN module
	LTE_IRAT/13/MultiLayer_Procedures_EG_GERAN.ttcn

	MCC160 Comment
	Alternative solution implemented instead: moved the creation of the TCH from the beginning of the test case to after the call to f_GERAN_InterRAT_InitialiseAuthParams.  This change made in: 13.1.7, 13.1.8, 13.1.9, 13.1.19, 13.1.20


Before change

...
  /*

   * @desc      REFERENCE TS 36.523-1 clause 13.1.8

   * @status    APPROVED (LTE_IRAT)

   */

  function f_TC_13_1_8_GERAN() runs on GERAN_PTC

  {

    var IRAT_Coordination_MSG v_OctetData;

    //Initialise cell info and configure cell 24

    f_GERAN_Init(EUTRA_GERAN);

    f_GERAN_CreateCell (geran_Cell24);

    f_GERAN_CreateTCH (geran_Cell24);
    //Set cell power according to "T0"

    f_IRAT_WaitForCoOrd_Trigger (EUTRA);

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_SuitableNeighbourCell);

    f_GERAN_SetChPwrLevel (geran_Cell24, tsc_TchCh, tsc_GERAN_SuitableNeighbourCell);
    f_GERAN_TestBody_Set(true);

    //Receive updated auth parameters

    f_GERAN_InterRAT_InitialiseAuthParams();

    EUTRA.receive (cmr_IRAT_OctetData) -> value v_OctetData; // @sic R5s130883 sic@

    //@siclog "Step 4-35" siclog@

    //Steps 1 to 32 of the generic test procedure described in TS36.508 subclause 6.4.3.8.2 are performed on Cell24

    f_GERAN_CSFB_Redirection_CCO_M0(geran_Cell24, oct2int(v_OctetData.OctetData)); // @sic R5s130883 sic@

    f_GERAN_PreliminaryPass (__FILE__, __LINE__, "Test Case 13.1.8 Step 35");

    //Tell EUTRA PTC that test case is over

    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger);

    f_GERAN_TestBody_Set(false);

    //Perform postamble

    f_GERAN_Postamble(geran_Cell24, G1_IDLE, -, -, tsc_TCH_F); // @sic R5s120500 sic@

  }

  /*

   * @desc      REFERENCE TS 36.523-1 clause 13.1.9

   * @status    APPROVED (LTE_IRAT)

   */

  function f_TC_13_1_9_GERAN() runs on GERAN_PTC

  { /* Call setup from E-UTRA RRC_IDLE / CS fallback to GSM with CCO without NACC / MO call */

    var IRAT_Coordination_MSG v_OctetData;

    //Initialise cell info and configure cell 24

    f_GERAN_Init(EUTRA_GERAN);

    f_GERAN_CreateCell (geran_Cell24);

    //Send cell info to EUTRA PTC

    f_GERAN_Send_IR_SysInfoToEUTRA (geran_Cell24);

    //Receive updated auth parameters

    f_GERAN_InterRAT_InitialiseAuthParams();

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_SuitableNeighbourCell);

    f_GERAN_CreateTCH (geran_Cell24);

    f_GERAN_SetChPwrLevel (geran_Cell24, tsc_TchCh, tsc_GERAN_SuitableNeighbourCell);
    f_GERAN_TestBody_Set(true);

    //@siclog "Step 8 Void" siclog@

    f_GERAN_InterRAT_InitialiseAuthParams(); // @sic R5s140002 sic@

    EUTRA.receive (cmr_IRAT_OctetData) -> value v_OctetData; // @sic R5s130883 sic@

    //@siclog "Step 9-40" siclog@

    f_GERAN_CSFB_Redirection_CCO_M0(geran_Cell24, oct2int(v_OctetData.OctetData));  // @sic R5s120699, R5s130883 sic@

    //Tell EUTRA PTC that test case is over

    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger);

    f_GERAN_TestBody_Set(false);

    //Perform postamble

    f_GERAN_Postamble(geran_Cell24, G1_IDLE, -, -, tsc_TCH_F); // @sic R5s120504 sic@

  }
...
After change

...

  /*

   * @desc      REFERENCE TS 36.523-1 clause 13.1.8

   * @status    APPROVED (LTE_IRAT)

   */

  function f_TC_13_1_8_GERAN() runs on GERAN_PTC

  {

    var IRAT_Coordination_MSG v_OctetData;

    //Initialise cell info and configure cell 24

    f_GERAN_Init(EUTRA_GERAN);

    f_GERAN_CreateCell (geran_Cell24);

    //Set cell power according to "T0"

    f_IRAT_WaitForCoOrd_Trigger (EUTRA);

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_SuitableNeighbourCell);

    f_GERAN_TestBody_Set(true);

    //Receive updated auth parameters

    f_GERAN_InterRAT_InitialiseAuthParams();

    EUTRA.receive (cmr_IRAT_OctetData) -> value v_OctetData; // @sic R5s130883 sic@

    //@siclog "Step 4-35" siclog@

    //Steps 1 to 32 of the generic test procedure described in TS36.508 subclause 6.4.3.8.2 are performed on Cell24

    f_GERAN_CSFB_Redirection_CCO_M0(geran_Cell24, oct2int(v_OctetData.OctetData)); // @sic R5s130883 sic@

    f_GERAN_PreliminaryPass (__FILE__, __LINE__, "Test Case 13.1.8 Step 35");

    //Tell EUTRA PTC that test case is over

    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger);

    f_GERAN_TestBody_Set(false);

    //Perform postamble

    f_GERAN_Postamble(geran_Cell24, G1_IDLE, -, -, tsc_TCH_F); // @sic R5s120500 sic@

  }

  /*

   * @desc      REFERENCE TS 36.523-1 clause 13.1.9

   * @status    APPROVED (LTE_IRAT)

   */

  function f_TC_13_1_9_GERAN() runs on GERAN_PTC

  { /* Call setup from E-UTRA RRC_IDLE / CS fallback to GSM with CCO without NACC / MO call */

    var IRAT_Coordination_MSG v_OctetData;

    //Initialise cell info and configure cell 24

    f_GERAN_Init(EUTRA_GERAN);

    f_GERAN_CreateCell (geran_Cell24);

    //Send cell info to EUTRA PTC

    f_GERAN_Send_IR_SysInfoToEUTRA (geran_Cell24);

    //Receive updated auth parameters

    f_GERAN_InterRAT_InitialiseAuthParams();

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_SuitableNeighbourCell);

    f_GERAN_TestBody_Set(true);

    //@siclog "Step 8 Void" siclog@

    f_GERAN_InterRAT_InitialiseAuthParams(); // @sic R5s140002 sic@

    EUTRA.receive (cmr_IRAT_OctetData) -> value v_OctetData; // @sic R5s130883 sic@

    //@siclog "Step 9-40" siclog@

    f_GERAN_CSFB_Redirection_CCO_M0(geran_Cell24, oct2int(v_OctetData.OctetData));  // @sic R5s120699, R5s130883 sic@

    //Tell EUTRA PTC that test case is over

    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger);

    f_GERAN_TestBody_Set(false);

    //Perform postamble

    f_GERAN_Postamble(geran_Cell24, G1_IDLE, -, -, tsc_TCH_F); // @sic R5s120504 sic@

  }
...
MCC160 Implementation:
  function f_TC_13_1_7_GERAN() runs on GERAN_PTC

  {

    var IRAT_Coordination_MSG v_OctetData;

    //Initialise cell info and configure cell 24

    f_GERAN_Init(EUTRA_GERAN);

    f_GERAN_CreateCell (geran_Cell24);

    
    //Send cell info to EUTRA PTC

    f_GERAN_Send_IR_SysInfoToEUTRA (geran_Cell24);

    //Receive updated auth parameters

    f_GERAN_InterRAT_InitialiseAuthParams();
    f_GERAN_CreateTCH (geran_Cell24);
    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_SuitableNeighbourCell); // @sic R5-123126 sic@

    f_GERAN_SetChPwrLevel (geran_Cell24, tsc_TchCh, tsc_GERAN_SuitableNeighbourCell); // @sic R5-123126 sic@

    f_GERAN_InterRAT_InitialiseAuthParams(); // @sic R5s140002 sic@

    EUTRA.receive (cmr_IRAT_OctetData) -> value v_OctetData; // @sic R5s130883 sic@

    f_GERAN_TestBody_Set(true);

    //@siclog "Step 7-40" siclog@

    //Steps 1 to 34 of the generic test procedure described in TS36.508 subclause 6.4.3.8.1 are performed on Cell24

    f_GERAN_CSFB_Redirection_CCO_MT(geran_Cell24, oct2int(v_OctetData.OctetData)); //@sic R5s130883 sic@

    f_GERAN_PreliminaryPass (__FILE__, __LINE__, "Test Case 13.1.7 Step 40");

    //Tell EUTRA PTC that test case is over

    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger);

    f_GERAN_TestBody_Set(false);

    //Perform postamble

    f_GERAN_Postamble(geran_Cell24, G1_IDLE, -, -, tsc_TCH_F); // @sic R5s120498 sic@

  }

  /*

   * @desc      REFERENCE TS 36.523-1 clause 13.1.8

   * @status    APPROVED (LTE_IRAT)

   */

  function f_TC_13_1_8_GERAN() runs on GERAN_PTC

  {

    var IRAT_Coordination_MSG v_OctetData;

    //Initialise cell info and configure cell 24

    f_GERAN_Init(EUTRA_GERAN);

    f_GERAN_CreateCell (geran_Cell24);


    //Set cell power according to "T0"

    f_IRAT_WaitForCoOrd_Trigger (EUTRA);

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_SuitableNeighbourCell);


    f_GERAN_TestBody_Set(true);

    //Receive updated auth parameters

    f_GERAN_InterRAT_InitialiseAuthParams();

    f_GERAN_CreateTCH (geran_Cell24);

    f_GERAN_SetChPwrLevel (geran_Cell24, tsc_TchCh, tsc_GERAN_SuitableNeighbourCell);

    EUTRA.receive (cmr_IRAT_OctetData) -> value v_OctetData; // @sic R5s130883 sic@

    //@siclog "Step 4-35" siclog@

    //Steps 1 to 32 of the generic test procedure described in TS36.508 subclause 6.4.3.8.2 are performed on Cell24

    f_GERAN_CSFB_Redirection_CCO_M0(geran_Cell24, oct2int(v_OctetData.OctetData)); // @sic R5s130883 sic@

    f_GERAN_PreliminaryPass (__FILE__, __LINE__, "Test Case 13.1.8 Step 35");

    //Tell EUTRA PTC that test case is over

    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger);

    f_GERAN_TestBody_Set(false);

    //Perform postamble

    f_GERAN_Postamble(geran_Cell24, G1_IDLE, -, -, tsc_TCH_F); // @sic R5s120500 sic@

  }

  /*

   * @desc      REFERENCE TS 36.523-1 clause 13.1.9

   * @status    APPROVED (LTE_IRAT)

   */

  function f_TC_13_1_9_GERAN() runs on GERAN_PTC

  { /* Call setup from E-UTRA RRC_IDLE / CS fallback to GSM with CCO without NACC / MO call */

    var IRAT_Coordination_MSG v_OctetData;

    //Initialise cell info and configure cell 24

    f_GERAN_Init(EUTRA_GERAN);

    f_GERAN_CreateCell (geran_Cell24);

    //Send cell info to EUTRA PTC

    f_GERAN_Send_IR_SysInfoToEUTRA (geran_Cell24);

    //Receive updated auth parameters

    f_GERAN_InterRAT_InitialiseAuthParams();

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_SuitableNeighbourCell);



    f_GERAN_TestBody_Set(true);

    //@siclog "Step 8 Void" siclog@

    f_GERAN_InterRAT_InitialiseAuthParams(); // @sic R5s140002 sic@

    f_GERAN_CreateTCH (geran_Cell24);

    f_GERAN_SetChPwrLevel (geran_Cell24, tsc_TchCh, tsc_GERAN_SuitableNeighbourCell);

    EUTRA.receive (cmr_IRAT_OctetData) -> value v_OctetData; // @sic R5s130883 sic@

    //@siclog "Step 9-40" siclog@

    f_GERAN_CSFB_Redirection_CCO_M0(geran_Cell24, oct2int(v_OctetData.OctetData));  // @sic R5s120699, R5s130883 sic@

    //Tell EUTRA PTC that test case is over

    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger);

    f_GERAN_TestBody_Set(false);

    //Perform postamble

    f_GERAN_Postamble(geran_Cell24, G1_IDLE, -, -, tsc_TCH_F); // @sic R5s120504 sic@

  }

  function f_TC_13_1_19_20_GERAN_Common() runs on GERAN_PTC

  {

    var G_L2_DATAMESSAGE_IND v_ServReq;

    var G_L2_DATAMESSAGE_IND v_SetUp;

    var SubChannelNumber v_SubChannel;

    var G_LogicChType v_ChannelType;

    var G_LogicChType v_TchType;

    var TI v_Ti_R;

    var RFN v_RFN;

    var integer v_RR_RA;

    var G_L2_DATAMESSAGE_IND v_ChanReq;

    var B10_Type v_BCCH := f_GERAN_BCCHFreq_Get(geran_Cell24);

    var B3_Type v_BCC := f_GERAN_BCC_Get(geran_Cell24);

    timer t_Wait1s := 1.0;

    //Initialise cell info and configure cell 24

    f_GERAN_Init(EUTRA_GERAN);

    f_GERAN_Send_IR_SysInfoToEUTRA(geran_Cell24);

    f_GERAN_Send_IR_SysInfoToEUTRA(geran_Cell25);

    f_GERAN_Send_IR_SysInfoToEUTRA(geran_Cell26);

    f_GERAN_CreateCell(geran_Cell24);


    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    f_GERAN_SetCellPowerLevel(geran_Cell24, tsc_GERAN_SuitableNeighbourCell);


    f_GERAN_TestBody_Set(true);

    //@siclog "Step 5b1" siclog@

    f_GERAN_InterRAT_InitialiseAuthParams(); // @sic R5s160065 sic@

    f_GERAN_CreateTCH(geran_Cell24);

    f_GERAN_SetCellPowerLevel(geran_Cell24, tsc_GERAN_ServingCell);

    f_GERAN_SetChPwrLevel(geran_Cell24, tsc_TchCh, tsc_GERAN_ServingCell);

    ...

